Monte-Carlo Integration

Contents

1 Monte-Carlo Integration 2
1.1 One Dimension 2
1.2 The Central Limit Theorem 3
1.3 Estimating Volumes 5
1.4 Multiple Integrals 6
1 Monte-Carlo Integration

1.1 One Dimension

Suppose we wish to estimate an integral

\[ I = \int_{a}^{b} f(x)dx \]

The average value of \( f(x) \) is

\[ \bar{f} = \frac{1}{b-a}I \]

so that

\[ I = (b-a)\bar{f} \]

If we choose points \( x_1, \ldots, x_n \) randomly in the interval \([a, b]\) and use these to compute the average value, \( \bar{f} \), of \( f(x) \), then

\[ I \approx (b-a)\bar{f} \]

This is Monte-Carlo integration.

Example

Let us estimate the integral

\[ I = \int_{0}^{2\pi} e^{-x}\sin(x)dx \]

The exact value, from integration by parts, is

\[ I = \frac{1}{2}(1 - e^{-2\pi}) = 0.4990663. \]

Here is a Monte-Carlo estimate using Scilab:

```function i = monte(n)
    x = 2*pi*rand(1,n);
    fx = exp(-x).*sin(x);
    i = 2*pi*sum(fx)/n
endfunction```
-->ii = monte(100000)
ii =

0.4983886

We can compare to the exact answer and relative error:

-->ie = (1-exp(-2*pi))/2
ie =

0.4990663

-->err = abs(ii-ie)/ie
err =

0.001357

i.e. we get a relative error of the order of 0.1%.

Let us see how the error varies with \( n \):

-->m = 1:18;

-->n = 2.^m;

-->in = zeros(1,18);

-->for k = 1:18
   --> in(k) = monte(n(k));
-->end

-->err = abs(in - ie)/ie;

-->plot2d(m, log10(err))

1.2 The Central Limit Theorem

The central limit theorem of probability theory gives an estimate of the error in Monte-Carlo integration. For our purposes it can be formulated as follows:
Figure 1: Error in Monte-Carlo Integration

Suppose the mean of a function \( f(x) \) is estimated by random sampling

\[
\bar{f}_{\text{est}} = \frac{1}{n} \sum_{i=1}^{n} f(x_i)
\]

Then the variance of the estimated mean is

\[
\text{Var} \bar{f} = \frac{\sigma^2}{n}
\]

where \( \sigma^2 \) is the variance of \( f(x) \).

It is usual to take the standard deviation as a measure of error. Then this says that

\[
\text{Error} = \frac{\sigma}{\sqrt{n}}
\]

The important point here is that the error goes to zero like \( 1/\sqrt{n} \). This says, for example, that to decrease the error by a factor of 1000, we must increase the sample size by a factor of 1000000.
We will compare our results in the previous example with this theory. The variance of \( f(x) \) is

\[
\sigma^2 = \int_0^{2\pi} (f(x) - \bar{f})^2 \approx 1.19
\]

so the expected error in Monte-Carlo integration is

\[
E_n = \frac{\sigma}{\sqrt{n}} \approx \frac{1.09}{\sqrt{n}}
\]

\( \text{--->ee} = 1.09*(n).^(-1/2); \)

\( \text{--->plot2d(m', [log10(err)', log10(ee)']);} \)

1.3 Estimating Volumes

What is the volume of a unit sphere in 4-dimensions? We can obtain an estimate by Monte-Carlo methods.
The unit sphere is the region
\[ x_1^2 + x_2^2 + x_3^2 + x_4^2 \leq 1 \]
If we generate random points in the four dimensional cubic region \([-1,1] \times [-1,1] \times [-1,1] \times [-1,1]\) which contains the unit sphere, then an estimate of the volume of the sphere is:

\[
\frac{\text{Volume of Sphere}}{\text{Volume of Cube}} \approx \frac{\text{No. Points in Sphere}}{\text{Total No. of Points}}
\]

By symmetry we get the same result if work in a single quadrant, say \([0,1] \times [0,1] \times [0,1] \times [0,1]\).

Here is the calculation in Scilab.

```scilab
function v = mvol(n)
    k = 0    
    // count no. points in sphere
    for i = 1:n
        x = rand(1,4)
        if (norm(x) <= 1) then
            k = k + 1
        end
    end
    v = 16*k/n    
    // 16 quadrants!
endfunction
```

```octave
-->mvol(10000)
ans  =

4.9248
```

By the way, the exact answer is \( \pi^2/2 = 4.9348 \)

### 1.4 Multiple Integrals

For one-dimensional integration problems Monte-Carlo integration is quite inefficient. For high-dimensional integration it is a useful technique. The reasons for this are twofold:

1. The fact that error is proportional to \(1/\sqrt{n}\) does not depend on the dimension. In other words it performs just as well in high dimensions as in one dimension.
2. It can easily handle regions with irregular boundaries. The method used in the previous section to estimate volumes can easily be adapted to Monte-Carlo integration over any region.

**Example**

We will evaluate the integral

\[
I = \int_{\Omega} \sin \sqrt{\ln(x + y + 1)} \, dx \, dy
\]

where \( \Omega \) is the disk

\[
\left( x - \frac{1}{2} \right)^2 + \left( y - \frac{1}{2} \right)^2 \leq \frac{1}{4}
\]

```matlab
function i = monte2(n)
    k = 0
    sumf = 0
    while (k < n)
        x = rand(1,1)
        y = rand(1,1)
        if ((x-0.5)^2 + (y-0.5)^2 <= 0.25) then // (x,y) is in disk
            k = k + 1
            sumf = sumf + sin(sqrt(log(x+y+1)))
        end
    end
    i = (%pi/4)*(sumf/n) // %pi/4 = volume of disk
endfunction
```

```matlab
-->monte2(10000)
ans =

0.5697654
```